SPATIAL EQUILIBRIUM MODELS WITH GAMS

ERWIN KALVELAGEN

Abstract. This document gives a short introduction into Spatial Equilibrium modeling with GAMS.

1. Intertemporal-Spatial Price Equilibrium

Solving spatial equilibrium models by mathematical programming is pioneered by [9],[12], [11], and [13].

One of the most basic forms is the transportation problem, also known as the “Koopmans-Hitchcock” model. An example is found in the GAMS model library in `trnsport.gms`.

2. Spatial Price Equilibrium

Here we will develop a Spatial and Temporal Price and Allocation (STPA) model. We assume that a (linear) demand and supply curve is available [4]:

$$\begin{aligned} d_j &= \alpha_j + \beta_j p_j \\ s_i &= \gamma_i + \delta_i \pi_i \end{aligned} \tag{1}$$

with $\beta_j < 0$, $\delta_i > 0$. The indices i and j are used to denote supply and demand regions. p and π are demand and supply prices. Inverse formulations of the demand and supply equations are as follows:

$$\begin{aligned} p_j &= \zeta_j + \eta_j d_j \\ \pi_i &= \theta_i + \lambda_i s_i \end{aligned} \tag{2}$$

with $\zeta_j, \theta_i, \lambda_i > 0$ and $\eta_j < 0$. In addition we have standard constraints on supply and demand quantities with respect to transportation quantities $x_{i,j} \geq 0$:

$$\begin{aligned} \sum_{i=1}^{n} x_{i,j} &\geq d_j \\ \sum_{j=1}^{n} x_{i,j} &\leq s_i \end{aligned} \tag{3}$$

We assume further that unit transportation costs are denoted by $c_{i,j} \geq 0$. Then the objective is to maximize the global sum of producers' and consumers' surplus after deduction of the transportation costs. This objective is called the “net quasi-welfare function” [13] or “net social payoff” [9] and is defined by:

$$\max \sum_{j=1}^{n} \int_0^{d_j} d_j(\phi_j) d\phi_j - \sum_{i=1}^{n} \int_0^{s_i} \pi_i(\varphi_i) d\varphi_i - \sum_{i=1}^{n} \sum_{j=1}^{n} c_{i,j} x_{i,j} \tag{4}$$

Date: March 12, 2003.

If we integrate the functions 2, the following quadratic objective will result:

$$\max \sum_{j=1}^{n} \left(\zeta_j d_j + \frac{1}{2} \eta_j d_j^2 \right) - \sum_{i=1}^{n} \left(\theta_i s_i + \frac{1}{2} \lambda_i s_i^2 \right) - \sum_{i=1}^{n} \sum_{j=1}^{n} c_{i,j} x_{i,j} \tag{5}$$

A small illustrative numerical example from [13] can be implemented in GAMS as follows:

2.1. Model speq1.gms. [1]

```
$ontext

   Simple Spatial Equilibrium Example
   Primal or quantity formulation.

   Erwin Kalvelagen, nov 2001

   Data from:
      Takayama and Judge,
      Spatial and temporal price and allocation models
      North Holland, 1971

$offtext

set i 'regions' /region1,region2,region3/;
alias (i,j);

positive variables
    d(j)   'demand'
    s(i)   'supply'
    pd(j)  'price (demand)'
    ps(i)  'price (supply)'
    x(i,j) 'shipments'
;
free variable
    welfare   'quasi welfare'
;
equations
    excess_demand(j)  'no excess demand allowed'
    excess_supply(i)  'excess supply is possible'
    objective         'net quasi welfare function'
    price_demand(j)   'demand curve (inverse)'
    price_supply(i)   'supply curve (inverse)'
;

table data(i,*,*) 'data for inverse demand and supply equations'
                  intercept  slope
region1.demand      20       -0.1
region1.supply       5        0.1
region2.demand      20       -0.2
region2.supply       2.5      0.05
region3.demand      20       -0.125
region3.supply       5        0.1
;

table c(i,j) 'transportation costs'
         region1  region2  region3
region1              2        2
region2     2                 1
region3     2        1
;

parameters zeta(j),eta(j),theta(i),lambda(i);
zeta(j)   = data(j,'demand','intercept');
eta(j)    = data(j,'demand','slope');
theta(i)  = data(i,'supply','intercept');
lambda(i) = data(i,'supply','slope');
```

[1] http://www.gams.com/~erwin/micro/speq1.gms

```
price_demand(j).. pd(j) =e= zeta(j)+eta(j)*d(j);
price_supply(i).. ps(i) =e= theta(i)+lambda(i)*s(i);

objective.. welfare =e=
     sum(j,zeta(j)*d(j)+0.5*eta(j)*sqr(d(j)))
    -sum(i,theta(i)*s(i)+0.5*lambda(i)*sqr(s(i)))
    -sum((i,j), c(i,j)*x(i,j));

excess_demand(j).. sum(i, x(i,j)) =g= d(j);
excess_supply(i).. sum(j, x(i,j)) =l= s(i);

model m /objective,excess_demand,excess_supply,price_demand,price_supply/;
solve m using nlp maximizing welfare;
```

The equations `price_demand` and `price_supply` are not really needed, as they only are used to calculate the prices. Such equations are called "accounting rows". The same effect can be achieved by an assignment statement after the `SOLVE` statement.

A different formulation is used in [15]. The equations 3 are replaced by a single equation:

$$D_i - \sum_{j=1}^{n}(x_{j,i} - x_{i,j}) \leq s_i \tag{6}$$

I.e. it is assumed that there is transportation cost involved if a good is produced and consumed in the same region, i.e. $x_{i,i} = 0$.

3. Complementarity formulation

The model assumes inverse demand and supply functions are available and are integrable. These conditions are not always met, in which case different formulations can be devised. The previous section showed a primal or quantity or Marshallian formulation of the problem, but dual (also known as price or Walrasian formulation) and complementarity formulations are available [10, 13, 3, 15]. To be complete we also mention that [13] present a primal-dual formulation that does not require integrability.

The KKT conditions resulting in the complementarity formulation are especially interesting as they have a direct economic interpretation. Because we have a quadratic model, the Karush-Kuhn-Tucker conditions lead to a linear complementarity model.

Let the Lagrangean be:

$$\begin{aligned}\mathcal{L}(d,s,x,\mu,\nu) = &\sum_{j=1}^{n}\int_0^{d_j} p_j(\phi_j)d\phi_j - \sum_{i=1}^{n}\int_0^{s_i}\pi_i(\varphi_i)d\varphi_i - \sum_{i=1}^{n}\sum_{j=1}^{n}c_{i,j}x_{i,j}\\ &- \sum_{j=1}^{n}\mu_j\left(\sum_{i=1}^{n}x_{i,j} - d_j\right) - \sum_{i=1}^{n}\nu_i\left(\sum_{j=1}^{n}x_{i,j} - s_i\right)\end{aligned} \tag{7}$$

This results, by setting partial derivatives to zero, in the following first order conditions:

- No excess demand and efficient market pricing:

$$\sum_i x_{i,j} \geq d_j \perp p_j \geq 0 \tag{8}$$

i.e. if the demand price in region j is positive, all demand will be met, otherwise the demand price will be zero.

- Excess supply is possible and efficient demand pricing:

$$\sum_j x_{i,j} \leq s_i \perp \pi_i \geq 0 \tag{9}$$

i.e. if the supply price in region i is positive, there will be no more produced than needed, otherwise the supply price is zero.

- Spatial price equilibrium:

$$\pi_i + c_{i,j} \geq p_j \perp x_{i,j} \geq 0 \tag{10}$$

i.e. there will only be goods transported from i to j if the difference in supply and demand price in region i and j is large enough to cover the transportation cost $c_{i,j}$. Shippers will not make profits: the best they can hope for is to recover their costs. If they would make a profit, another shipper would enter the market and start offering lower prices.

These conditions form an LCP which we can code in GAMS.

3.1. **Model speq2.gms.** [2]

```
$ontext

   Simple Spatial Equilibrium Example
   Complementarity formulation

   Erwin Kalvelagen, nov 2001

   Data from:
      Takayama and Judge,
      Spatial and temporal price and allocation models
      North Holland, 1971


$offtext

set i 'regions' /region1,region2,region3/;
alias (i,j);

positive variables
    pd(j)  'price (demand)'
    ps(i)  'price (supply)'
    x(i,j) 'shipments'
;
free variable
    d(j)   'demand'
    s(i)   'supply'
;
equations
    excess_demand(j)    'no excess demand allowed'
    excess_supply(i)    'excess supply is possible'
    transportation(i,j) 'transportation costs'
    demand(j)           'demand curve'
    supply(i)           'supply curve'
;

table data(i,*,*)
                 intercept   slope
region1.demand      200      -10
region1.supply      -50       10
region2.demand      100       -5
region2.supply      -50       20
region3.demand      160       -8
region3.supply      -50       10
```

[2] http://www.gams.com/~erwin/micro/speq2.gms

```
;

table c(i,j) 'transportation costs'
         region1  region2  region3
region1            2        2
region2     2               1
region3     2        1
;

parameters alpha(j),beta(j),gamma(i),delta(i);
alpha(j) = data(j,'demand','intercept');
beta(j)  = data(j,'demand','slope');
gamma(i) = data(i,'supply','intercept');
delta(i) = data(i,'supply','slope');

demand(j).. d(j) =e= alpha(j)+beta(j)*pd(j);
supply(i).. s(i) =e= gamma(i)+delta(i)*ps(i);

excess_demand(j)..  sum(i, x(i,j)) =g= d(j);
excess_supply(i)..  s(i) =g= sum(j, x(i,j));

transportation(i,j)..  ps(i) - pd(j) + c(i,j) =g= 0;

option mcp=path;
model m /demand.d,supply.s,excess_demand.pd,excess_supply.ps,transportation.x/;
solve m using mcp;
```

Note that we have included the demand and supply equations (1). In this case there is no need to form the inverse demand and supply equations.

4. Price formulation

The price or dual formulation is stated in terms of prices instead of quantities:

$$
\begin{aligned}
&\min \sum_{i=1}^{n} \int_0^{\pi_i} s_i(\rho_i) d\rho_i - \sum_{j=1}^{n} \int_0^{p_j} d_j(\sigma_j) d\sigma_j \\
&\pi_i + c_{i,j} \geq p_j \\
&\pi_i, p_j \geq 0
\end{aligned}
\tag{11}
$$

Using the demand and supply equations (1), we can rewrite the objective function as:

$$
\sum_{i=1}^{n} \left(\gamma_i \pi_i + \frac{1}{2} \delta_i (\pi_i)^2 \right) - \sum_{j=1}^{n} \left(\alpha_j p_j + \frac{1}{2} \beta_j (p_j)^2 \right)
\tag{12}
$$

The model below implements this dual formulation:

4.1. Model speq3.gms. [3]

```
$ontext

   Simple Spatial Equilibrium Example
   Dual or price formulation

   Erwin Kalvelagen, nov 2001

   Data from:
      Takayama and Judge,
      Spatial and temporal price and allocation models
      North Holland, 1971


$offtext
```

[3] http://www.gams.com/~erwin/micro/speq3.gms

```
set i 'regions' /region1,region2,region3/;
alias (i,j);

positive variables
    pd(j)  'price (demand)'
    ps(i)  'price (supply)'
;

free variable z 'objective variable';

equations
    transcosts(i,j) 'transportation costs'
    objective
;

table data(i,*,*)
                intercept   slope
region1.demand      200      -10
region1.supply      -50       10
region2.demand      100       -5
region2.supply      -50       20
region3.demand      160       -8
region3.supply      -50       10
;

table c(i,j) 'transportation costs'
         region1  region2  region3
region1              2        2
region2     2                 1
region3     2        1
;

parameters alpha(j),beta(j),gamma(i),delta(i);
alpha(j) = data(j,'demand','intercept');
beta(j)  = data(j,'demand','slope');
gamma(i) = data(i,'supply','intercept');
delta(i) = data(i,'supply','slope');


objective..        z =e= sum(i, gamma(i)*ps(i) + 0.5*delta(i)*sqr(ps(i)))
                        -sum(j, alpha(j)*pd(j) + 0.5*beta(j)*sqr(pd(j)));
transcosts(i,j)..  ps(i) - pd(j) + c(i,j) =g= 0;

model m /objective, transcosts/;
solve m using nlp minimizing z;

parameters
    d(j) 'demand'
    s(i) 'supply'
;
d(j) = alpha(j)+beta(j)*pd.l(j);
s(i) = gamma(i)+delta(i)*ps.l(i);

display d,s;

parameter x(i,j) 'quantities transported';
x(i,j) = transcosts.m(i,j);
display x;
```

The demand and supply quantities can be calculated using the equations (1), and the shipping quantities $x_{i,j}$ are the dual variables of the equation $\pi_i + c_{i,j} \geq p_j$.

5. Multi-period models

Direct extensions of the above models include handling multiple commodities, multiple time periods and storage of goods.

For the multi-commodity, multi-period case, the linear demand and supply functions will look like:

$$
\begin{aligned}
D_{i,k,t} &= \alpha_{i,k,t} + \sum_{\ell} \beta_{i,k,\ell,t} P^D_{i,\ell,t} \\
S_{j,k,t} &= \gamma_{j,k,t} + \sum_{\ell} \delta_{i,k,\ell,t} P^S_{i,\ell,t}
\end{aligned} \tag{13}
$$

Inverting these equations, yield:

$$
\begin{aligned}
P^D_{i,k,t} &= \zeta_{i,k,t} + \sum_{\ell} \eta_{i,k,\ell,t} D_{i,\ell,t} \\
P^S_{j,k,t} &= \theta_{j,k,t} + \sum_{\ell} \lambda_{j,k,\ell,t} S_{i,\ell,t}
\end{aligned} \tag{14}
$$

If we allow storage of products, denoted by $I_{i,k,t,t+1}$ with associated costs $d_{i,k,t,t+1}$ then the following equilibrium conditions can be formulated:

- No excess demand and efficient market pricing:

$$
\sum_i x_{i,j,k,t} + I_{i,k,t-1,t} \geq D_{j,k,t} \perp P^D_{j,k,t} \geq 0 \tag{15}
$$

In order for the integral in (4) to be well-defined the Jacobian of the supply and demand functions need to be symmetric, i.e. $\eta_{i,k,\ell,t} = \eta_{i,\ell,k,t}$ and $\lambda_{j,k,\ell,t} = \lambda_{j,\ell,k,t}$. In many cases this is not a viable assumption, either from economic point of view or for statistical reasons. Here the complementarity formulation shows its strength: it does not require the symmetry conditions.

In addition to the symmetry conditions, there is also the need for the objective function to be concave (as we are maximizing). This means that the matrices $B^t_i = \beta_{i,k,\ell,t}$ and $\Lambda^t_j = \lambda_{j,k,\ell,t}$ are supposed to be positive definite.

6. Price controls

An example of a policy intervention is price control. Adding a price floor on the supply price or a price ceiling on the demand price is quite simple [7].

Let $\bar{p}_j$ be the demand price ceiling and $\underline{\pi}_i$ the supply price floor. We introduce explicit excess supply u_i and excess demand v_j as follows:

$$
\begin{aligned}
d_j &= \sum_i x_{i,j} + v_j \\
s_i &= \sum_j x_{i,j} + u_i
\end{aligned} \tag{16}
$$

A complete complementarity formulation would look like:

$$
\begin{aligned}
d_j &= \sum_i x_{i,j} + v_j \perp d_j \text{ free} \\
s_i &= \sum_j x_{i,j} + u_i \perp s_i \text{ free} \\
\pi_i + c_{i,j} &\geq p_j \perp x_{i,j} \geq 0 \\
\pi_i &\geq \underline{\pi}_i \perp u_i \geq 0 \\
p_j &\leq \bar{p}_j \perp v_j \geq 0
\end{aligned} \tag{17}
$$

Real world examples of implementations of a spatial equilibrium can be found in [14], [5], [17], [16], [1], [8] and [2]. The text [6] has a chapter on price endogeneous models which includes much what is discussed here, in addition to an example in GAMS.

References

1. Chi-Chung Chen, Bruce A. McCarl, and Ching-Cheng Chang, *Spatial equilibrium modeling with imperfectly competitive markets: An application to rice trade*, `http://ageco.tamu.edu/faculty/mccarl/papers/736.pdf`.
2. Steven A. Gabriel, Julio Manik, and Shree Vikas, *Computational experience with a large-scale, multi-period, spatial equilibrium model of the north american natural gas system*, Tech. report, University of Maryland, 2001.
3. P. T. Harker, *Alternative models of spatial competition*, Operations Research **34** (1986), no. 3, 410–425.
4. Walter C. Labys, *Modeling mineral and energy markets*, Kluwer Academic Publishers, 1999.
5. B. A. McCarl, C. Chang, J. D. Atwood, and W. I. Nayda, *Documentation of ASM: the U.S. Agricultural Sector Model*, Tech. report, Texas A&M University, College Station, TX, 1993.
6. B. A. McCarl and T. H. Spreen, *Applied mathematical programming using algebraic systems*, `http://agecon.tamu.edu/faculty/mccarl/regbook.htm`, 1996.
7. Anna Nagurney, *Network economics, a variational inequality approach*, revised second ed., Advances in Computational Economics, vol. 10, Kluwer Academic Publishers, 1999.
8. A. Ruijs, C. Schweigman, C. Lutz, and G. Sirpé, *Cereal trade in developing countries: stochastic spatial equilibrium models*, Tech. report, University of Groningen, 2001.
9. P. Samuelson, *Spatial price equilibrium and linear programming*, American Economic Review **42** (1952), 283–303.
10. T. Takayama and A. D. Woodland, *Equivalence of price and quantity formulations of spatial equilibrium: Purified duality in quadratic and concave programming*, Econometrica **38** (1970), no. 6, 889–906.
11. Takashi Takayama and George B. Judge, *An intertemporal price equilibrium model*, American Journal of Agricultural Economics (1964), 477–484.
12. ———, *Spatial equilibrium and quadratic programming*, American Journal of Agricultural Economics (1964), 67–93.
13. ———, *Spatial and temporal price and allocation models*, North-Holland, 1971.
14. ———, *Studies in economic planning over space and time*, North-Holland, 1973.
15. Sten Thore, *Economic logistics: the optimization of spatial and sectoral resource, production and distribution systems*, Quorum Books, 1991.
16. J. C. J. M. van den Bergh, P. Nijkamp, and P. Rietveld, *Recent advances in spatial equilibrium modelling: methodology and applications*, Springer Verlag, 1995.
17. Paulo D. Waquil and Thomas L. Cox, *Spatial equilibrium with intermediate products: Implementation and validation in the MERCOSUR*, Tech. Report 388, University of Wisconsin-Madison, Agricultural Economics, December 1995.

GAMS Development Corp., Washington DC
E-mail address: `erwin@gams.com`